Problem:
Score : 500 points

Problem Statement

There are N towns on a plane. The i-th town is located at the coordinates (x_i,y_i). There may be more than one town at the same coordinates.

You can build a road between two towns at coordinates (a,b) and (c,d) for a cost of min(|a-c|,|b-d|) yen (the currency of Japan). It is not possible to build other types of roads.

Your objective is to build roads so that it will be possible to travel between every pair of towns by traversing roads. At least how much money is necessary to achieve this?

Constraints

2 ≤ N ≤ 10^5

0 ≤ x_i,y_i ≤ 10^9

All input values are integers.

Input

Input is given from Standard Input in the following format:

N
x_1 y_1
x_2 y_2
:
x_N y_N

Output

Print the minimum necessary amount of money in order to build roads so that it will be possible to travel between every pair of towns by traversing roads.

Sample Input 1

3
1 5
3 9
7 8

Sample Output 1

3

Build a road between Towns 1 and 2, and another between Towns 2 and 3. The total cost is 2+1=3 yen.

Sample Input 2

6
8 3
4 9
12 19
18 1
13 5
7 6

Sample Output 2

8

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

{-# LANGUAGE FlexibleContexts #-}

import Data.Array.ST
import Data.Array
import Data.List
import Data.Ord
import qualified Data.ByteString.Char8 as BS
import Data.Maybe
import Control.Monad
import Control.Monad.ST
import Control.Monad.State
import Control.Applicative

type UnionFindST s = STArray s Int (Int,Int)

newUFST :: Int -> ST s (UnionFindST s)
newUFST n = newListArray (1,n) (zip [1..n] (repeat 1))

findUFST :: Int -> UnionFindST s -> ST s Int
findUFST x uf = do
  (parent,size) <- readArray uf x
  if parent == x then
    return x
    else do
    next <- findUFST parent uf
    writeArray uf x (next,size)
    return next

sameUFST :: Int -> Int -> UnionFindST s -> ST s Bool
sameUFST x y uf = do
  xx <- findUFST x uf
  yy <- findUFST y uf
  return $ xx == yy

uniteUFST :: Int -> Int -> UnionFindST s -> ST s ()
uniteUFST x y uf = do
  xx <- findUFST x uf
  yy <- findUFST y uf
  (_,xs) <- readArray uf xx
  (_,ys) <- readArray uf yy

  if xx == yy then
    return ()
    else if xs < ys then do
    writeArray uf xx (yy,0)
    else if xs > ys then do
    writeArray uf yy (xx,0)
    else do
    writeArray uf yy (xx,0)
    writeArray uf xx (xx,xs+1)

type Node = Int
type Cost = Int
type Edge = (Node, Node, Cost)

kruskal :: [Node] -> [Edge] -> [Edge]
kruskal nodes edges = runST $ do
  let sorted = sortBy (\(_,_,c1) (_,_,c2) -> compare c1 c2) edges
  uf <- newUFST (length nodes)
  execStateT (kruskal' uf sorted) []
  where
    kruskal' _ [] = return ()
    kruskal' uf (e@(i,j,c):rest) = do
      isSame <- lift $ sameUFST i j uf
      unless isSame $ do
        lift $ uniteUFST i j uf
        modify (e:)
      kruskal' uf rest

listToTuple2 (a:b:_) = (a,b)

main = do
  n <- readLn :: IO Int
  nodes <- zip [1..] . map (listToTuple2 . map (fst . fromJust . BS.readInt) . BS.words) . BS.lines <$> BS.getContents :: IO [(Int,(Int,Int))]

  
  let roads = map (\((n1,d1),(n2,d2)) -> (n1,n2,dist d1 d2)) $ f (sortBy (comparing (fst . snd)) nodes) ++ f (sortBy (comparing (snd . snd)) nodes)

  print $ sum . map (\(_,_,x) -> x) $ kruskal [1..n] roads

  where
    f (_:[]) = []
    f (x:y:xs) = (x,y) : f (y:xs)
    dist (x1,y1) (x2,y2) = min (abs $ x1-x2) (abs $ y1-y2)